AD-A178 616 COMPUTER PROGRAM FOR ELECTROMAGNETIC PENETRATION INTO A CONDUCTING CIRCUL..(U) SYRACUSE UNIV NY DEPT OF ELECTRICAL AND COMPUTER ENGINEERING.. J R MAUTZ ET AL. FEB 87 SYRU/DECE/TR87/1 ARO-21378.9-EL 1/1

UNCLASSIFIED F/G 20/14 NL




1.0
1.1
1.25
1.4
1.6
1.8
2.0
2.2
2.8
3.2
3.6
4.0

DTIC
ELECTE
S D
A

Approved for public release; distribution unlimited

Prepared for

ARMY RESEARCH OFFICE
RESEARCH TRIANGLE PARK
NORTH CAROLINA 27709

87 4 1 306

SYRU/DECE/TR-87/1

COMPUTER PROGRAM FOR ELECTROMAGNETIC PENETRATION INTO A CONDUCTING CIRCULAR CYLINDER THROUGH A NARROW SLOT, TM CASE

Interim Technical Report No. 7

by

Joseph R. Mautz

Roger F. Harrington

February 1987

Department of
Electrical and Computer Engineering
Syracuse University
Syracuse, New York 13244-1240

Contract No. DAAG29-84-K-0078

Prepared for

ARMY RESEARCH OFFICE
RESEARCH TRIANGLE PARK
NORTH CAROLINA 27709

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER
ARO 21378.9-EL

2. GOVT ACCESSION NO.
ADA178616

3. RECIPIENT'S CATALOG NUMBER

4. TITLE (and Subtitle)
COMPUTER PROGRAM FOR ELECTROMAGNETIC PENETRATION INTO A CONDUCTING CIRCULAR CYLINDER THROUGH A NARROW SLOT, TM CASE

5. TYPE OF REPORT & PERIOD COVERED
Interim Technical Report #7

6. PERFORMING ORG. REPORT NUMBER

7. AUTHOR(s)
Joseph R. Mautz
Roger F. Harrington

8. CONTRACT OR GRANT NUMBER(s)
DAAG29-84-K-0078

9. PERFORMING ORGANIZATION NAME AND ADDRESS
Department of Electrical & Computer Engineering
Syracuse University
Syracuse, New York 13244-1240

10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS

11. CONTROLLING OFFICE NAME AND ADDRESS
U.S. Army Research Office
Post Office Box 12211
Research Triangle Park, NC 27709

12. REPORT DATE
February 1987

13. NUMBER OF PAGES
42

14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)

15. SECURITY CLASS. (of this report)
UNCLASSIFIED

15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

NA

18. SUPPLEMENTARY NOTES

The view, opinions, and/or findings contained in this report are those of the author(s) and should not be construed as an official Department of the Army position, policy, or decision, unless so designated by other documentation.

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Aperture problems
Circular cylinder
Electromagnetic penetration
Computer program
Numerical solution
Resonance

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A computer program is described and listed. This program calculates electromagnetic penetration of a TM plane wave into the cylindrical cavity for which $\rho \leq a$ when the surface $(\rho = a, \cancel{\phi_0 \leq \phi \leq 2\pi - \phi_0})$ is perfectly conducting. Here, ρ and ϕ are cylindrical coordinates. The z directed electric field E_z in the slot aperture $(\rho = a, \cancel{-\phi_0 \leq \phi} \leq \phi_0)$ is expressed as a linear combination of 4 even and 4 odd expansion functions. The method of moments is used to obtain the coefficients of these functions. The elements of the

DD FORM 1473, 1 JAN 73 — EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-014-6601

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

20. ABSTRACT (continued)

(phi sub 0)

moment matrix are obtained by expressing each expansion function as a Fourier series in ϕ valid for $(0 \leq \phi \leq 2\pi)$.

Our moment solution remains accurate when the cavity becomes resonant because alternative expansion functions were chosen to prevent the moment matrix from becoming ill-behaved. As the aperture width $2\phi_0$ decreases, more and more terms in the Fourier series are needed. Thanks to Debye's asymptotic expansions for Bessel functions, we are able to handle 10,000 terms so as to obtain accurate results for ϕ_0 as small as 1.25°.

phi sub 0 deg

CONTENTS

INTRODUCTION

In [1], a procedure is presented for calculating the electromagnetic field in the vicinity of an infinitesimally thin perfectly conducting circular cylindrical shell with an infinitely long slot illuminated by a TM wave. A computer program was written to implement this procedure. For the shell of [1, Fig. 1] illuminated by the incident electric field $\underline{E}^i$ given by [1, eq. (1)]

$$\underline{E}^i = \underline{u}_z e^{jk\rho \cos(\phi-\alpha)} , \tag{1}$$

the program calculates the amplitude $|E_z|$ of the z directed electric field at equally spaced points in the slot aperture and along the straight line that runs from the center of the cylinder to the center of the aperture. The radius of the shell is a, and the angular width of the slot aperture is $2\phi_o$. The incident electric field is the electric field that would exist if the shell were absent. In (1), (ρ,ϕ,z) are cylindrical coordinates, and $\underline{u}_z$ is the unit vector in the z-direction. The incident wave (1) comes from the direction for which $\phi = \alpha$. In (1), $k = \omega\sqrt{\mu\varepsilon}$ where ω is the angular frequency, μ is the permeability of the homogeneous medium that surrounds the shell, and ε is the permittivity of this medium.

The computer program consists of the subroutines BES, BESJ, BESJY, BESJJ, SFEO, DECOMP, and SOLVE, and a main program. The main program reads input data on the file MAUTZ3.DAT and writes output data on file MAUTZ6.DAT. The user who wants merely to run the program is advised to skip to Section VIII and to read only the descriptions of the input and output data there.

II. THE SUBROUTINE BES

The subroutine BES(YØ, N, X, BJ, BY) puts the Bessel function of the first kind $J_n(X)$ in BJ(n+1) and the Bessel function of the second kind $Y_n(X)$ in BY(n+1) for n=0,1,...,N where X is a positive real number. The subroutine BES differs from that in [2, Sec. II] only in that common input variables YØ(1) to YØ(33), PI2, PI4, PI7, in [2, p. 6] have been changed to the subroutine arguments YØ(1) to YØ(33), YØ(34), YØ(35), and YØ(36), respectively, and the common output variables BJ and BY have been changed to the subroutine arguments BJ and BY, respectively. The symbol Ø in YØ denotes zero rather than the letter O. In other instances when it is obvious from the context that a zero is meant, the slash is not used.

Minimum allocations are given by

DIMENSION BJ(N+1), BY(N+1), AJ(11+N+2*[X]) where [X] is the largest integer that does not exceed X.

```
C     LISTING OF THE SUBROUTINE BES
      SUBROUTINE BES(YO,N,X,BJ,BY)
      DIMENSION YO(36),BJ(100),BY(100),AJ(100)
      MZ=10+N+2*IFIX(X)
      IF(X.LT.1.E-3) MZ=4+N
      AJ(MZ+1)=0.
      AJ(MZ)=1.E-20
      M1=MZ-1
      X2=2./X
      DO 16 K=1,M1
      MK=MZ-K
      AJ(MK)=X2*FLOAT(MK)*AJ(MK+1)-AJ(MK+2)
   16 CONTINUE
      ALP=.5*AJ(1)
      DO 15 J=3,MZ,2
      ALP=ALP+AJ(J)
   15 CONTINUE
      ALP=2.*ALP
      NP=N+1
      DO 17 K=1,NP
      BJ(K)=AJ(K)/ALP
   17 CONTINUE
      IF(X-8.) 18,18,19
   18 Z1=X*X
      Z2=0.
      Z3=0.
      Z4=0.
      Z5=0.
      Z6=0.
      Z7=0.
      IF(Z1.LT.1.E-11) GO TO 22
      Z2=Z1*Z1
      Z3=Z2*Z1
      IF(Z1.LT.1.E-6) GO TO 22
      Z4=Z3*Z1
      Z5=Z4*Z1
      IF(Z1.LT.1.E-4) GO TO 22
      Z6=Z5*Z1
      Z7=Z6*Z1
   22 S=(YO(1)+YO(2)*Z1+YO(3)*Z2+YO(4)*Z3+YO(5)*Z4+YO(6)*Z5+
     1YO(7)*Z6+YO(8)*Z7)/(YO(9)+YO(10)*Z1+YO(11)*Z2)
      TLG=ALOG(X)
      BY(1)=S+YO(34)*BJ(1)*TLG
      IF(N.LE.0) RETURN
      S=(YO(12)+YO(13)*Z1+YO(14)*Z2+YO(15)*Z3+YO(16)*Z4+
     1YO(17)*Z5+YO(18)*Z6)/(YO(19)+YO(20)*Z1+YO(21)*Z2)
      BY(2)=X*S+YO(34)*(BJ(2)*TLG-1./X)
      GO TO 20
   19 Z=8./X
      Z1=Z*Z
      Z2=Z1*Z1
      S=SQRT(YO(34)/X)
      S1=YO(22)+YO(23)*Z1+YO(24)*Z2
      S3=Z*(YO(28)+YO(29)*Z1+YO(30)*Z2)
      S5=X-YO(35)
      BY(1)=S*(S1*SIN(S5)+S3*COS(S5))
      IF(N.LE.0) RETURN
      S2=YO(25)+YO(26)*Z1+YO(27)*Z2
      S4=Z*(YO(31)+YO(32)*Z1+YO(33)*Z2)
      S6=X-YO(36)
      BY(2)=S*(S2*SIN(S6)+S4*COS(S6))
   20 IF(NP.LE.2) RETURN
      DO 21 K=3,NP
      K2=K-2
      BY(K)=X2*FLOAT(K2)*BY(K-1)-BY(K2)
   21 CONTINUE
      RETURN
      END
```

III. THE SUBROUTINE BESJ

The subroutine BESJ(X, BJZ, BJ1) puts $J_o(X)$ in BJZ and $J_1(X)$ in BJ1 where X is a positive real number.

If $X \leq 3$, then $J_o(X)$ and $J_1(X)$ are calculated in lines 5 to 20 in the listing of BESJ at the end of this section. As suggested in [3, Sec. 9.12., Example 1], $J_{MZ}(X)$ and $J_{MZ-1}(X)$ are set equal to the arbitrary values of zero and 10^{-20}, respectively, where

$$MZ = 11 + 2[X] \tag{2}$$

where [X] is the largest integer that does not exceed X. This is done in lines 5 to 7. In DO loop 16, the recurrence relation [3, Eq. (9.1.27)]

$$J_n(X) = \frac{2(n+1)}{X} J_{n+1}(X) - J_{n+2}(X) \tag{3}$$

is used to calculate $\{J_n(X),\ n = MZ-2, MZ-3,\ldots,0\}$. Line 12 puts $J_{MK-1}(X)$ in BJ(MK). According to [3, eq. (9.1.46)], the calculated values of $J_o(X)$ and $J_1(X)$ have to be normalized by dividing by α where

$$\alpha = J_o(X) + 2J_2(X) + 2J_4(X) + \ldots \tag{4}$$

The division by α is performed in lines 19 and 20.

If $X > 3$, then $J_o(X)$ and $J_1(X)$ are calculated in lines 22 to 38. The polynomial approximations [3, Eqs. 9.4.3 and 9.4.6] are used. With regard to [3, Eq. 9.4.3], lines 28 and 29 put f_o in FZ, lines 30 and 31 put θ_o in TZ, and line 37 puts $J_o(X)$ in BJZ. With regard to [3, Eq. 9.4.6], lines 32 and 33 put f_1 in F1, lines 34 and 35 put θ_1 in T1, and line 38 puts $J_1(X)$ in BJ1.

```
C     LISTING OF THE SUBROUTINE BESJ
      SUBROUTINE BESJ(X,BJZ,BJ1)
      DIMENSION BJ(18)
      IF(X-3.) 17,17,18
   17 MZ=11+2*IFIX(X)
      BJ(MZ+1)=0.
      BJ(MZ)=1.E-20
      M1=MZ-1
      X2=2./X
      DO 16 K=1,M1
      MK=MZ-K
      BJ(MK)=X2*FLOAT(MK)*BJ(MK+1)-BJ(MK+2)
   16 CONTINUE
      ALP=.5*BJ(1)
      DO 19 J=3,MZ,2
      ALP=ALP+BJ(J)
   19 CONTINUE
      ALP=2.*ALP
      BJZ=BJ(1)/ALP
      BJ1=BJ(2)/ALP
      RETURN
   18 X1=3./X
      X2=X1*X1
      X3=X2*X1
      X4=X3*X1
      X5=X4*X1
      X6=X5*X1
      FZ=.7978846-.77E-6*X1-.55274E-2*X2-.9512E-4*X3+
     1.137237E-2*X4-.72805E-3*X5+.14476E-3*X6
      TZ=X-.7853982-.4166397E-1*X1-.3954E-4*X2+.262573E-2*X3-
     1.54125E-3*X4-.29333E-3*X5+.13558E-3*X6
      F1=.7978846+.156E-05*X1+.1659667E-01*X2+.17105E-03*X3-
     1.249511E-02*X4+.113653E-02*X5-.20033E-03*X6
      T1=X-2.356194+.1249961*X1+.565E-04*X2-.637879E-02*X3+
     1.74348E-03*X4+.79824E-03*X5-.29166E-03*X6
      XX=SQRT(X)
      BJZ=FZ/XX*COS(TZ)
      BJ1=F1/XX*COS(T1)
      RETURN
      END
```

IV. THE SUBROUTINE BESJY

The subroutine BESJY(N, X, BJY) puts $J_N(X)\ Y_N(X)$ in BJY where J_N and Y_N are, respectively, the Bessel functions of the first and second kinds of order N. Here, X is a non-negative real number, and N is large enough so that Debye's asymptotic expansions [3, eqs. (9.3.7) and (9.3.8)] apply to $J_N(X)$ and $Y_N(X)$. The subroutine BESJY is that in [2, Sec. VI] with DO loop 11 removed, the arguments N1 and N2 replaced by N, and the array BJY replaced by the single variable BJY.

```
C     LISTING OF THE SUBROUTINE BESJY
      SUBROUTINE BESJY(N,X,BJY)
      F1=N
      F2=F1*F1
      F3=F2*F1
      F4=F3*F1
      XN=X/F1
      T1=1./SQRT(1.-XN*XN)
      T2=T1*T1
      T3=T2*T1
      T4=T3*T1
      T5=T4*T1
      T6=T5*T1
      T7=T6*T1
      T8=T7*T1
      T9=T8*T1
      T10=T9*T1
      T12=T10*T2
      U1=(3.*T1-5.*T3)/(24.*F1)
      U2=(81.*T2-462.*T4+385.*T6)/(1152.*F2)
      U3=(30375.*T3-369603.*T5+765765.*T7-425425.*T9)/(414720.*F3)
      U4=(4465125.*T4-9412168.E+01*T6+3499224.E+02*T8-4461857.E+02*
     1T10+1859107.E+02*T12)/(3981312.E+01*F4)
      U5=1.+U2+U4
      U6=U1+U3
      BJY=-T1/(3.141593*F1)*(U5+U6)*(U5-U6)
      RETURN
      END
```

V. THE SUBROUTINE BESJJ

The subroutine BESJJ(N, NR, X, BJJ) puts $\frac{J_N(X(L))}{J_N(X(NR))}$ in BJJ(L) for L = 1,2,...,NR. Here, $X(L) \geq 0$ and N is large enough so that Debye's asymptotic expansion applies to J_N. Minimum allocations are given by

DIMENSION X(NR), BJJ(NR), U5(NR), U6(NR)

Consider $J_N(XX)$ where

$$XX = X(L) \tag{5}$$

Debye's asymptotic expansion for $J_N(XX)$ is [3, eq. (9.3.7)]

$$J_N(XX) = \frac{e^{N(\tanh\alpha-\alpha)}}{\sqrt{2\pi N \tanh\alpha}} \left(1 + \sum_{k=1}^{4} \frac{u_k(\coth\alpha)}{N^k}\right) \tag{6}$$

where u_k is given by [3, formula 9.3.9] and

$$\tanh\alpha = \sqrt{1 - \left(\frac{XX}{N}\right)^2} \tag{7}$$

From [4, formula 702.], we have

$$\alpha = \frac{1}{2} \ln \left(\frac{1 + \tanh\alpha}{1 - \tanh\alpha}\right) \tag{8}$$

Substituting (7) into (8) and rationalizing the denominator of the argument of the logarithm, we obtain

$$\alpha = \ln \left[\frac{N}{XX} (1 + \tanh\alpha)\right] \tag{9}$$

Substitution of (9) into (6) gives

$$J_N(XX) = \frac{1}{\sqrt{2\pi N \tanh\alpha}} \left(\frac{XX\ e^{\tanh\alpha}}{N(1+\tanh\alpha)}\right)^N \left(1 + \sum_{k=1}^{4} \frac{u_k(\coth\alpha)}{N^k}\right) \tag{10}$$

Thanks to (10), we obtain

$$\frac{J_N(XX)}{J_N(X(NR))} = \left(\frac{U6(L)}{U6(NR)}\right) \left(\frac{U5(L)}{U5(NR)}\right)^N \tag{11}$$

where

$$U5(L) = \frac{XX\ e^{\tanh\,\alpha}}{1 + \tanh\,\alpha} \tag{12}$$

and

$$U6(L) = \frac{1 + \sum_{k=1}^{4} \frac{u_k(\coth\,\alpha)}{N^k}}{\sqrt{\tanh\,\alpha}} \tag{13}$$

Here, $\tanh\alpha$ depends on L by means of (7) with XX given by (5).

In the listing of the subroutine BESJJ at the end of this section, line 5 inside DO loop 11 obtains XX of (5). Lines 23 to 26 put u_1, u_2, u_3, and u_4 of (13) in U1, U2, U3, and U4, respectively. Line 28 obtains U5(L) of (12). Line 29 obtains U6(L) of (13). Inside DO loop 12, line 34 puts $J_N(XX)/J_N(X(NR))$ of (11) in BJJ(L).

```
C     LISTING OF THE SUBROUTINE BESJJ
      SUBROUTINE BESJJ(N,NR,X,BJJ)
      DIMENSION X(11),BJJ(11),U5(11),U6(11)
      DO 11 L=1,NR
      XX=X(L)
      F1=N
      F2=F1*F1
      F3=F2*F1
      F4=F3*F1
      XN=XX/F1
      TT=SQRT(1.-XN*XN)
      T1=1./TT
      T2=T1*T1
      T3=T2*T1
      T4=T3*T1
      T5=T4*T1
      T6=T5*T1
      T7=T6*T1
      T8=T7*T1
      T9=T8*T1
      T10=T9*T1
      T12=T10*T2
      U1=(3.*T1-5.*T3)/(24.*F1)
      U2=(81.*T2-462.*T4+385.*T6)/(1152.*F2)
      U3=(30375.*T3-369603.*T5+765765.*T7-425425.*T9)/(414720.*F3)
      U4=(4465125.*T4-9412168.E+01*T6+3499224.E+02*T8-4461857.E+02*
     1T10+1859107.E+02*T12)/(3981312.E+01*F4)
      U5(L)=XX*EXP(TT)/(1.+TT)
      U6(L)=(1.+U1+U2+U3+U4)/SQRT(TT)
   11 CONTINUE
      S5=U5(NR)
      S6=U6(NR)
      DO 12 L=1,NR
      BJJ(L)=U6(L)/S6*(U5(L)/S5)**N
   12 CONTINUE
      BJJ(NR)=1.
      RETURN
      END
```

VI. THE SUBROUTINE SFEO

The subroutine SFEO(N, P, SE, SO, FE, FO) puts S^e_{JN} of [1, eqs. (A-18) to (A-21)] or [1, eqs. (A-32) to (A-35)] in SE(J), S^o_{JN} of [1, eqs. (A-22) to (A-25)] or [1, eqs. (A-37) to (A-40)] in SO(J), F^e_{JN} of [1, eqs. (B-5) to (B-8)] or [1, eqs. (B-18) to (B-21)] in FE(J), and F^o_{JN} of [1, eqs. (B-9) to (B-12)] or [1, eqs. (B-22) to (B-25)] in FO(J), all for J=1,2,3, and 4. The variables N and P are input arguments defined by

$$\left.\begin{aligned} N &= n \\ P &= \phi_o \end{aligned}\right\} \tag{14}$$

where n and ϕ_o appear in [1, eq. (B-4)] so that [1, eq. (B-4)] becomes

$$b = N*P \tag{15}$$

The subroutine SFEO calls the subroutine BESJ.

The subroutine SFEO is listed at the end of this section. If $b > 2$, line 11 of this listing puts $J_o(b)$ and $J_1(b)$ in BJZ and BJ1, respectively. Lines 15 to 19 put S^e_{JN} of [1, eqs. (A-18) to (A-21)] in SE(J) for J=1,2,3, and 4. Lines 22 to 26 put S^o_{JN} of [1, eqs. (A-22) to (A-25)] in SO(J) for J=1,2,3, and 4. Lines 30 to 33 put F^e_{JN} of [1, eqs. (B-5) to (B-8)] in FE(J) for J=1,2,3, and 4. Lines 36 to 40 put F^o_{JN} of [1, eqs. (B-9) to (B-12)] in FO(J) for J=1,2,3, and 4.

If $b \leq 2$, lines 42 and 43 put $\epsilon_n\phi_o/2$ in PP. Lines 46 to 53 put S^e_{JN} of [1, eqs. (A-32) to (A-35)] in SE(J) for J=1,2,3, and 4. Lines 55 to 62 put S^o_{JN} of [1, eqs. (A-37) to (A-40)] in SO(J) for J=1,2,3, and 4. Lines 63 to 70 put F^e_{JN} of [1, eqs. (B-18) to (B-21)] in FE(J) for J=1,2,3, and 4. Lines 71 to 78 put F^o_{JN} of [1, eqs. (B-22) to (B-25)] in FO(J) for J=1,2,3, and 4.

```
C     LISTING OF THE SUBROUTINE SFEO
C     THE SUBROUTINE SFEO CALLS THE SUBROUTINE BESJ.
      SUBROUTINE SFEO(N,P,SE,SO,FE,FO)
      DIMENSION SE(4),SO(4),FE(4),FO(4)
      FN=N
      B=FN*P
      B2=B*B
      B4=B2*B2
      B6=B4*B2
      IF(B.LE.2.) GO TO 11
      CALL BESJ(B,BJZ,BJ1)
      SZ=BJZ/FN
      BZ=B*SZ
      B1=BJ1/FN
      SE(1)=B1
      SE(2)=((B2-6.)*B1+3.*BZ)/B2
      SE(3)=((B4-27.*B2+120.)*B1+6.*(B2-10.)*BZ)/B4
      SE(4)=((B6-63.*B4+1200.*B2-5040.)*B1+3.*(3.*B4-95.*B2+840.)*B
     1Z)/B6
      BZ=SZ
      B1=B1/B
      SO(1)=2.*B1-BZ
      SO(2)=((5.*B2-24.)*B1-(B2-12.)*BZ)/B2
      SO(3)=((8.*B4-168.*B2+720.)*B1-(B4-39.*B2+360.)*BZ)/B4
      SO(4)=((11.*B6-537.*B4+9720.*B2-40320.)*B1-(B6-81.*B4+2340.*B
     12-20160.)*BZ)/B6
      SN=SIN(B)/FN
      SS=COS(B)/FN
      CS=B*SS
      FE(1)=SN
      FE(2)=(2.*CS+(B2-2.)*SN)/B2
      FE(3)=(4.*(B2-6.)*CS+(B4-12.*B2+24.)*SN)/B4
      FE(4)=(6.*(B4-20.*B2+120.)*CS+(B6-30.*B4+360.*B2-720.)*SN)/B6
      SN=SN/B
      CS=SS
      FO(1)=SN-CS
      FO(2)=(3.*(B2-2.)*SN-(B2-6.)*CS)/B2
      FO(3)=(5.*(B4-12.*B2+24.)*SN-(B4-20.*B2+120.)*CS)/B4
      FO(4)=(7.*(B6-30.*B4+360.*B2-720.)*SN-(B6-42.*B4+840.*B2-5040
     1.)*CS)/B6
      RETURN
   11 PP=P
      IF(N.EQ.0) PP=.5*PP
      B8=B6*B2
      B10=B8*B2
      SE(1)=PP/2.*(1.-B2/8.+B4/192.-B6/9216.+B8/737280.-B10/884736.
     1E+02)
      SE(2)=PP/8.*(1.-B2/4.+B4*5./384.-B6*7./23040.+B8/245760.-B10*
     111./3096576.E+02)
      SE(3)=PP/16.*(1.-B2*5./16.+B4*7./384.-B6*7./15360.+B8*11./172
     10320.-B10*143./2477261.E+03)
      SE(4)=PP*5./128.*(1.-B2*7./20.+B4*7./320.-B6*11./19200.+B8*14
     13./172032.E+02-B10*143./1857946.E+03)
```

```
      PB=P*B
      SO(1)=PB/8.*(1.-B2/12.+B4/384.-B6/23040.+B8/2211840.-B10/3096
     1576.E+02)
      SO(2)=PB/16.*(1.-B2*5./48.+B4*7./1920.-B6/15360.+B8*11./15482
     188.E+01-B10*13./2477261.E+03)
      SO(3)=PB*5./128.*(1.-B2*7./60.+B4*7./1600.-B6*11./134400.+B8*
     1143./1548288.E+02-B10*13./1857946.E+03)
      SO(4)=PB*7./256.*(1.-B2/8.+B4*11./2240.-B6*143/1505280.+B8*14
     13./1300562.E+02-B10*17./2000864.E+03)
      FE(1)=P*(1.-B2/6.+B4/120.-B6/5040.+B8/362880.-B10/399168.E+02
     1)
      FE(2)=P/3.*(1.-B2*.3+B4/56.-B6/2160.+B8/147840.-B10/157248.E+
     102)
      FE(3)=P/5.*(1.-B2*5./14.+B4*5./216.-B6/1584.+B8/104832.-B10/1
     108864.E+02)
      FE(4)=P/7.*(1.-B2*7./18.+B4*7./264.-B6*7./9360.+B8/86400.-B10
     1/8812800.)
      FO(1)=PB/3.*(1.-B2*.1+B4/280.-B6/15120.+B8/1330560.-B10/17297
     128.E+02)
      FO(2)=PB/5.*(1.-B2*5./42.+B4/216.-B6/11088.+B8/943488.-B10/11
     197504.E+02)
      FO(3)=PB/7.*(1.-B2*7./54.+B4*7./1320.-B6/9360.+B8/777600.-B10
     1/969408.E+02)
      FO(4)=PB/9.*(1.-B2*3./22.+B4*3./520.-B6/8400.+B8/685440.-B10/
     1842688.E+02)
       RETURN
       END
```

VII. THE SUBROUTINES DECOMP AND SOLVE

The subroutines DECOMP(N, IPS, UL) and SOLVE(N, IPS, UL, B, X) solve a system of N linear equations in N unknowns. The input to DECOMP consists of N and the N by N matrix of coefficients on the left-hand side of the matrix equation stored by columns in UL. The output from DECOMP is IPS and UL. This output is fed into solve. The rest of the input to solve consists of N and the column of coefficients on the right-hand side of the matrix equation stored in B. SOLVE puts the solution to the matrix equation in X.

Minimum allocations are given by

```
COMPLEX UL(N*N)
DIMENSION SCL(N), IPS(N)
```

in DECOMP and by

```
COMPLEX UL(N*N), B(N), X(N)
DIMENSION IPS(N)
```

in SOLVE

More detail concerning DECOMP and SOLVE is on pages 46-49 of [5].

```
C     LISTING OF THE SUBROUTINE DECOMP
      SUBROUTINE DECOMP(N,IPS,UL)
      COMPLEX UL(16),PIVOT,EM
      DIMENSION SCL(4),IPS(4)
      DO 5 I=1,N
      IPS(I)=I
      RN=0.
      J1=I
      DO 2 J=1,N
      ULM=ABS(REAL(UL(J1)))+ABS(AIMAG(UL(J1)))
      J1=J1+N
      IF(RN-ULM) 1,2,2
    1 RN=ULM
    2 CONTINUE
      SCL(I)=1./RN
    5 CONTINUE
      NM1=N-1
      K2=0
      DO 17 K=1,NM1
      BIG=0.
      DO 11 I=K,N
      IP=IPS(I)
      IPK=IP+K2
      SIZE=(ABS(REAL(UL(IPK)))+ABS(AIMAG(UL(IPK))))*SCL(IP)
      IF(SIZE-BIG) 11,11,10
   10 BIG=SIZE
      IPV=I
   11 CONTINUE
      IF(IPV-K) 14,15,14
   14 J=IPS(K)
      IPS(K)=IPS(IPV)
      IPS(IPV)=J
   15 KPP=IPS(K)+K2
      PIVOT=UL(KPP)
      KP1=K+1
      DO 16 I=KP1,N
      KP=KPP
      IP=IPS(I)+K2
      EM=-UL(IP)/PIVOT
   18 UL(IP)=-EM
      DO 16 J=KP1,N
      IP=IP+N
      KP=KP+N
      UL(IP)=UL(IP)+EM*UL(KP)
   16 CONTINUE
      K2=K2+N
   17 CONTINUE
      RETURN
      END
C     LISTING OF THE SUBROUTINE SOLVE
      SUBROUTINE SOLVE(N,IPS,UL,B,X)
      COMPLEX UL(16),B(4),X(4),SUM
      DIMENSION IPS(4)
      NP1=N+1
      IP=IPS(1)
      X(1)=B(IP)
      DO 2 I=2,N
      IP=IPS(I)
      IPB=IP
      IM1=I-1
      SUM=0.
      DO 1 J=1,IM1
      SUM=SUM+UL(IP)*X(J)
    1 IP=IP+N
    2 X(I)=B(IPB)-SUM
      K2=N*(N-1)
      IP=IPS(N)+K2
      X(N)=X(N)/UL(IP)
      DO 4 IBACK=2,N
      I=NP1-IBACK
      K2=K2-N
      IPI=IPS(I)+K2
      IP1=I+1
      SUM=0.
      IP=IPI
      DO 3 J=IP1,N
      IP=IP+N
    3 SUM=SUM+UL(IP)*X(J)
    4 X(I)=(X(I)-SUM)/UL(IPI)
      RETURN
      END
```

VIII. THE MAIN PROGRAM

The main program accepts input data from the file MAUTZ3.DAT and writes output data on the file MAUTZ6.DAT. See the open statements on lines 10 and 11 of the listing of the main program at the end of this section. The main program calls the subroutines BES, BESJY, BESJJ, SFEO, DECOMP, and SOLVE. The subroutine SFEO calls the subroutine BESJ. In this section, first the input data are described, then the output data are described, relevant formulas are given, the main program is described verbally, and finally the main program and sample input and output data are listed.

The input data are read early in the main program from the file MAUTZ3.DAT according to

```
      READ(20,25)(YØ(I), I = 1, 36)
25    FORMAT(5E14.7)
      READ(20,28) N1, N2, N3, NA, NR
28    FORMAT(I3, I5, 3I3)
      DO 31 JW = 1, N3
      READ(20,32) NN, IP, IB, X, P, ALP
32    FORMAT(3I3, 3E14.7)
31    CONTINUE
```

The input array YØ should not deeply concern the user because he will never have to change it. The values of the elements of YØ appear in the sample data listed after the main program. The curious reader will find the values of YØ(1) to YØ(33) explained on pages 4 and 5 of [2]. The values of YØ(34), YØ(35), and YØ(36) are $2/\pi$, $\pi/4$, and $3\pi/4$, respectively.

The Bessel functions in the summations in [1, eqs. (41), (42), (47), and (48)] are approximated by Debye's asymptotic expansions whenever $N1 \leq n \leq N2$, and these summations are truncated at $n = N2$. The summations in [1, eqs. (44) and (50)] are truncated at $n = N1-1$. For the z directed electric field $E_z^b(-\underline{M})$ along the straight line that runs from the center of the cylinder to the center of the aperture, the Bessel functions in the first summation in (62) are approximated by Debye's asymptotic expansions whenever $N1 \leq n \leq N2$, and this summation is truncated at $n = N2$.

The aperture field E_z is evaluated at NA equally spaced points in the aperture. Here, $NA \geq 2$. With reference to [1, Fig. 1], the first point is at the beginning of the aperture at $\phi = -\phi_o$, and the NAth point is at the end of the aperture at $\phi = \phi_o$. The interior field $E_z^b(-\underline{M})$ is evaluated at NR equally spaced points along the line that runs from the center of the cylinder to the center of the aperture. Here, $NR \geq 2$. The first point is at the center of the cylinder, and the NRth point is at the center of the aperture.

The NA evaluations of the aperture field and the NR evaluations of the interior field $E_z^b(-\underline{M})$ are done inside DO loop 31, that is, these evaluations are done N3 times, once for each value of JW. If $NN = 4$, these fields are obtained by solving [1, eqs. (40) and (46)] in which Y^e and Y^o are, as stated in [1], 4×4 matrices. If $1 \leq NN \leq 3$, the fields are obtained by solving [1, eqs. (40) and (46)] with Y^e replaced by the $NN \times NN$ matrix in its upper left-hand corner, with Y^o replaced by the $NN \times NN$ matrix in its upper left-hand corner, and with $\vec{V}^e$, $\vec{I}^e$, and $\vec{V}^o$, and $\vec{I}^o$ truncated accordingly. We assume that $1 \leq NN \leq 4$.

In DO loop 31, IP = p where p is the non-negative integer upon which the alternative expansion functions [1, eqs. (30) and (34)] depend. The main program was written under the assumption that IP < N1. If there is an integer p such that $|J_p(ka)|$ is very small, then the user should take IP = p. Here, k is the wave number and a is the radius of the cylinder. However, if there is no value of p such that $|J_p(ka)|$ is very small, a non-negative value of IP must still be chosen, and the alternative expansion functions will be given by [1, eqs. (30) and (34)] with p = IP. In this case the fields calculated by the procedure of [1] should not depend on p so that IP can be any non-negative integer less than N1. If IB ≠ 0, then the main program uses the value of $J_{IP}(ka)$ calculated by the subroutine BES. If IB = 0, then $J_{IP}(ka)$ is set equal to zero. If the circular cylinder of radius a was resonant such that $J_{IP}(ka) = 0$, then, without IB, it would be virtually impossible to obtain $J_{IP}(ka) = 0$ because the exact resonant value of ka would be difficult to enter, and even if it could be entered, the subroutine BES would probably not give exactly zero for $J_{IP}(ka)$.

In DO loop 31,

$$\left.\begin{aligned} IP &= p \\ X &= ka \\ P &= \phi_o \\ ALP &= \alpha \end{aligned}\right\} \qquad (16)$$

where p, ka, ϕ_o, and α appear in [1]. Here, P and ALP are in radians. The meaning of p was clarified in the previous paragraph, ka is the product of the wave number k and the radius a of the cylinder, ϕ_o is one half the angular width of the aperture, and α is the value of ϕ in the direc-

tion from which the incident wave comes. The angles ϕ_o and α are shown in [1, Fig. 1].

Minimum allocations are given by

```
DIMENSION BJ(MAX(N1*NR, NA)), BY(N1*NR), XR(NR), RP(NR)
DIMENSION RN(NN*NR), RE(NR), BJJ(NR)
```

where MAX denotes the maximum value.

The main program writes output data on the file MAUTZ6.DAT. The input data of the first two read statements are written out immediately after they are read in.

The following write statements are in DO loop 31:

```
      WRITE(21, 27) NN, IP, IB, X, P, ALP
27    FORMAT(' NN=', I3,', IP=', I3, ', IB = ', I3/
      1' X = ', E14.7, ', P=', E14.7, ', ALP=', E14.7)
      WRITE(21, 33) (BJ(NP), NP=JJ, JJN)
33    FORMAT(' BJ'/(1X, 5E14.7))
      WRITE(21, 19) (BY(NP), NP = JJ, JJN)
19    FORMAT(' BY'/(1X, 5E14.7))
      WRITE(21, 24) (SE(J), J=1, NN)
      WRITE(21, 24) (SO(J), J=1, NN)
24    FORMAT(1X, 4E14.7)
      WRITE(21, 49) U2, S1
49    FORMAT(' E=', 2E14.7,', ABS(E) = ', E14.7)
      WRITE(21,64) (BJ(L), L = 1, NA)
64    FORMAT(' APERTURE FIELD AMPLITUDE'/(1X, 4E14.7))
      WRITE(21, 60)(RE(L), L = 1, NR)
60    FORMAT(' INTERIOR FIELD AMPLITUDE'/(1X, 4E14.7))
```

The above write statements are labeled the first, second, third, ..., eighth write statements. The data of the first write statement are merely input data. The data of the second and third write statements are defined by

$$(BJ(NP),\ NP = JJ,\ JJN) = (J_n(ka),\ n=0,1,\ldots,\ N1-1) \tag{17}$$

$$(BY(NP),\ NP = JJ,\ JJN) = (Y_n(ka),\ n = 0,\ 1,\ldots,N1-1) \tag{18}$$

where $J_n(ka)$ and $Y_n(ka)$ are, respectively, the Bessel functions of the first and second kinds of order n and argument ka. The data of the fourth and fifth write statements are defined by

$$(SE(J),\ J=1,\ NN) = (|V_j^e|\ ,\ \ j=1,2,\ldots,NN) \tag{19}$$

$$(SO(J),\ J=1,\ NN) = (|V_j^o|\ ,\ \ j=1,2,\ldots,NN) \tag{20}$$

where V_j^e appears in [1, eq. (7)] and V_j^o appears in [1, eq. (24)]. If $NN < 4$, then V_j^e and V_j^o are truncated at $j = NN$ because they are given in terms of $\hat{V}_j^e$ and $\hat{V}_j^o$ by [1, eqs. (53) and (54)] and, as indicated earlier, $\hat{V}_j^e$ and $\hat{V}_j^o$ are truncated at $j = NN$.

In the sixth write statement, U2 is the complex number whose real and imaginary parts are those of $E_z^b(-\underline{M})$ at the center of the cylinder, and S1 is $|E_z^b(-\underline{M})|$ at the center of the cylinder. This $E_z^b(-\underline{M})$ is calculated from [1, eq. (68)]. In the seventh write statement, BJ(L) is the magnitude of the aperture field E_z at $\rho = a$ and $\phi = \phi_L$ where

$$\phi_L = \frac{2(L-1)\ \phi_o}{NA-1} - \phi_o \tag{21}$$

In the eighth write statement, RE(L) is $|E_z^b(-\underline{M})|$ at $k\rho = k\rho_L$ along the line that runs from the center of the cylinder to the center of the aperture. Here,

$$k\rho_L = \frac{(L - 1)\ ka}{NR-1} \tag{22}$$

This completes our description of the input and output data. The formulas to be programmed are presented next.

Assuming that

$$0 \leq p \leq N1 - 1, \tag{23}$$

we approximate the matrix elements [1, eqs. (41), (42), and (44)] by

$$Y^e_{i1} = \frac{F^e_{ip} S^e_{1p}}{H^{(2)}_p(ka)} + J_p(ka)\left[\sum_{\substack{n=0\\n\neq p}}^{N1-1} \frac{F^e_{in} S^e_{1n}}{J_n(ka) H^{(2)}_n(ka)} + j \sum_{n=N1}^{N2} \frac{F^e_{in} S^e_{1n}}{J_n(ka) Y_n(ka)}\right],$$

$$i=1,2,3,4 \tag{24}$$

$$Y^e_{ij} = \sum_{\substack{n=0\\n\neq p}}^{N1-1} \frac{F^e_{in}(S^e_{jn} + C^e_j S^e_{1n})}{J_n(ka)\ H^{(2)}_n(ka)} + j \sum_{n=N1}^{N2} \frac{F^e_{in}(S^e_{jn} + C^e_j S^e_{1n})}{J_n(ka)\ Y_n(ka)}, \quad \begin{cases} i=1,2,3,4 \\ j=2,3,4 \end{cases} \tag{25}$$

$$I^e_i = \sum_{n=0}^{N1-1} \frac{\varepsilon_n j^n F^e_{in} \cos(n\alpha)}{H^{(2)}_n(ka)}, \quad i=1,2,3,4 \tag{26}$$

In (25), C^e_j is given by [1, eq. (31)]

$$C^e_j = -\frac{S^e_{jp}}{S^e_{1p}} \tag{27}$$

Expressions (24) and (25) are recast as

$$Y^e_{i1} = S^e_{1p}(EF)_{ip} + J_p(ka) \sum_{\substack{n=0\\n\neq p}}^{N2} E^e_{1n}(EF)_{in}, \quad i=1,2,3,4 \tag{28}$$

$$Y^e_{ij} = \sum_{\substack{n=0 \\ n\neq p}}^{N2} E^e_{jn} (EF)_{in} \ , \quad \begin{cases} i=1,2,3,4 \\ j=2,3,4 \end{cases} \tag{29}$$

where

$$(EF)_{in} = \begin{cases} \dfrac{F^e_{in}}{H^{(2)}_n(ka)} \ , & n=0,1,\ldots,N1-1 \\ \\ j\,F^e_{in} \ , & n = N1,\ N1+1,\ldots,N2 \end{cases} \tag{30}$$

$$E^e_{jn} = \begin{cases} \dfrac{S^e_{jn} + C^e_j\, S^e_{1n}}{J_n(ka)} \ , & n=0,1,\ldots,p-1,p+1,\ldots,N1-1 \\ \\ \dfrac{S^e_{jn} + C^e_j\, S^e_{1n}}{J_n(ka)\ Y_n(ka)} \ , & n=N1,\ N1+1,\ldots,N2 \end{cases} \tag{31}$$

$$C^e_j = \begin{cases} 0 \ , & j=1 \\ \\ -\dfrac{S^e_{jp}}{S^e_{1p}} \ , & j=2,3,4 \end{cases} \tag{32}$$

Assuming that (23) holds, we approximate the matrix elements [1, eqs. (47), (48), and (50)] by

$$Y^o_{i1} = \frac{F^o_{ip}\ S^o_{1p}}{H^{(2)}_p(ka)} + J^o_p(ka)\left[\sum_{\substack{n=1 \\ n\neq p}}^{N1-1} \frac{F^o_{in}\ S^o_{1n}}{J_n(ka)\ H^{(2)}_n(ka)} + j\sum_{n=N1}^{N2} \frac{F^o_{in}\ S^o_{1n}}{J_n(ka)\ Y_n(ka)}\right] ,$$

$$i=1,2,3,4 \tag{33}$$

$$Y^o_{ij} = \sum_{\substack{n=1 \\ n\neq p}}^{N1-1} \frac{F^o_{in}(S^o_{jn}+C^o_j S^o_{1n})}{J_n(ka)H^{(2)}_n(ka)} + j\sum_{n=N1}^{N2} \frac{F^o_{in}(S^o_{jn}+C^o_j S^o_{1n})}{J_n(ka)\ Y_n(ka)} \ , \begin{cases} i=1,2,3,4 \\ j=2,3,4 \end{cases} \tag{34}$$

$$I_i^o = 2 \sum_{n=1}^{N1-1} \frac{j^n F_{in}^o \sin(n\alpha)}{H_n^{(2)}(ka)} \ , \ i=1,2,3,4 \tag{35}$$

where

$$J_p^o(ka) = \begin{cases} 1 & , \ p = 0 \\ J_p(ka) & , \ p > 0 \end{cases} \tag{36}$$

and [1, eq. (35)]

$$C_j^o = \begin{cases} 0 & , \ p = 0 \\ -\dfrac{S_{jp}^o}{S_{1p}^o} & , \ p > 0 \end{cases} \tag{37}$$

Expressions (33) and (34) are recast as

$$Y_{i1}^o = S_{1p}^o (OF)_{ip} + J_p^o(ka) \sum_{\substack{n=0 \\ n \neq p}}^{N2} E_{1n}^o (OF)_{in} \ , \ i=1,2,3,4 \tag{38}$$

$$Y_{ij}^o = \sum_{\substack{n=0 \\ n \neq p}}^{N2} E_{jn}^o (OF)_{in} \ , \ \begin{cases} i=1,2,3,4 \\ j=2,3,4 \end{cases} \tag{39}$$

where

$$(OF)_{in} = \begin{cases} \dfrac{F_{in}^o}{H_n^{(2)}(ka)} & , \ n=0,1,\ldots,N1-1 \\ j \ F_{in}^o & , \ n=N1,N1+1,\ldots,N2 \end{cases} \tag{40}$$

$$E_{jn}^o = \begin{cases} \dfrac{S_{jn}^o + C_j^o S_{1n}^o}{J_n(ka)} & , \ n=0,1,\ldots,p-1,p+1,\ldots,N1-1 \\ \dfrac{S_{jn}^o + C_j^o S_{1n}^o}{J_n(ka) Y_n(ka)} & , \ n=N1,N1+1,\ldots,N2 \end{cases} \tag{41}$$

$$C_j^o = \begin{cases} 0 & , \ p = 0,\ j=1,2,3,4 \\ 0 & , \ p > 0,\ j=1 \\ -\dfrac{S_{jp}^o}{S_{1p}^o} & , \ p > 0,\ j=2,3,4 \end{cases} \tag{42}$$

The terms for which n=0 do not contribute to (38) and (39) because $S_{jo}^o = F_{io}^o = 0$ [1, eqs. (27) and (49)]. The terms for which n=0 have been added to make (38) and (39) similar to (28) and (29) and thus easier to program.

For convenience, [1, eqs. (40), (46), and (53)] are repeated here:

$$Y^e \hat{\vec{V}}^e = \vec{I}^e \tag{43}$$

$$Y^o \hat{\vec{V}}^o = \vec{I}^o \tag{44}$$

$$\left.\begin{aligned} V_1^e &= J_p(ka)\ \hat{V}_1^e + \sum_{i=2}^{4} C_i^e \hat{V}_i^e \\ V_j^e &= \hat{V}_j^e\ , \quad j=2,3,4 \end{aligned}\right\} \tag{45}$$

With $J_p^o(ka)$ given by (36) and C_j^o by (42), [1, eqs. (54) and (55)] combine:

$$\left.\begin{aligned} V_1^o &= J_p^o(ka)\hat{V}_1^o + \sum_{i=2}^{4} C_i^o \hat{V}_i^o \\ V_j^o &= \hat{V}_j^o\ , \ j=2,3,4 \end{aligned}\right\} \tag{46}$$

Expression [1, eq. (68)] for the field $E_z^b(-\underline{M})$ at the center of the cylinder is

$$[E_z^b(-\underline{M})]_{\rho=0} = \begin{cases} S_{10}^e\ \hat{V}_1^e & , \ p = 0 \\ \dfrac{1}{J_o(ka)} \displaystyle\sum_{j=1}^{4} S_{jo}^e\ V_j^e & , \ p \neq 0 \end{cases} \tag{47}$$

The aperture field E_z is the ϕ component of $\underline{M}$ of [1, eq. (4)]:

$$E_z = M^e + M^o \tag{48}$$

Substituting [1, eqs. (7) and (24)] into (48), using [1, eqs. (8) and (25)], and evaluating E_z at $\phi = \{\phi_L, L = 1,2,\ldots,NA\}$ where ϕ_L is given by (21), we obtain

$$[E_z]_L = [\sum_{j=1}^{4} (\frac{\phi_L}{\phi_o} V_j^o + V_j^e) (\frac{\phi_L}{\phi_o})^{2j-2}] \sqrt{1 - (\frac{\phi_L}{\phi_o})^2}, \quad L=1,2,\ldots,NA \tag{49}$$

Along the line from the center of the cylinder to the center of the aperture, $\phi = 0$ so that expression [1, eq. (62)] for the interior field $E_z^b(-\underline{M})$ reduces to

$$E_z^b(-\underline{M}) = \sum_{n=0}^{\infty} \frac{B_n^e J_n(k\rho)}{J_n(ka)} \tag{50}$$

where B_n^e is given by [1, eqs. (63) or (65)].

$$B_n^e = \begin{cases} \sum_{j=1}^{4} V_j^e S_{jn}^e, & n \neq p \\ J_p(ka)\, \hat{V}_1^e S_{1p}^e, & n=p \end{cases} \tag{51}$$

Truncating the summation in (50) at n = N2, substituting (51) into (50), and then letting $k\rho = \{k\rho_L, L=1,2,\ldots,NR\}$ where $k\rho_L$ is given by (22), we have

$$[E_z^b(-\underline{M})]_L = S_{1p}^e J_p(k\rho_L)\hat{V}_1^e + \sum_{j=1}^{4} V_j^e (\sum_{\substack{n=0 \\ n\neq p}}^{N2} \frac{S_{jn}^e J_n(k\rho_L)}{J_n(ka)}), \quad L=1,2,\ldots,NR \tag{52}$$

Expression (52) is recast as

$$[E_z^b(-\underline{M})]_L = (RP)_L \hat{V}_1^e + \sum_{j=1}^{4} (RN)_{jL} V_j^e, \quad L = 1,2,\ldots,NR \tag{53}$$

where

$$(RP)_L = S^e_{1p}\; J_p(k\rho_L) \tag{54}$$

$$(RN)_{jL} = \sum_{\substack{n=0 \\ n\neq p}}^{N1-1} \frac{S^e_{jn}\; J_n(k\rho_L)}{J_n(ka)} + \sum_{n=N1}^{N2} \frac{S^e_{jn} J_n(k\rho_L)}{J_n(ka)} \tag{55}$$

Expression (55) was obtained by assuming that (23) holds. We plan to use Debye's asymptotic expansions for the Bessel functions in the second summation in (55).

If L=1, then $k\rho_L = 0$ and (53) reduces to

$$[E^b_z(-\underline{M})]_1 = \begin{cases} S^e_{10}\hat{V}^e_1 & , \; p = 0 \\ \sum\limits_{j=1}^{4} \left(\dfrac{S^e_{jo}}{J_o(ka)}\right) V^e_j & , \; p \neq 0 \end{cases} \tag{56}$$

Therefore, any difference between the computed values of the field (47) at the center of the cylinder and the field (53) at L = 1 is due to roundoff error.

If L = NR, then $k\rho_L = ka$ and (52) reduces to

$$[E^b_z(-\underline{M})]_{NR} = S^e_{1p}J_p(ka)\hat{V}^e_1 + \sum_{j=1}^{4} V^e_j \left(\sum_{\substack{n=0 \\ n\neq p}}^{N2} S^e_{jn}\right) \tag{57}$$

Using (45) to express $J_p(ka)\hat{V}^e_1$ in terms of $\{V^e_j,\; j=1,2,3,4\}$, we obtain

$$J_p(ka)\hat{V}^e_1 = V^e_1 - \sum_{i=2}^{4} C^e_i V^e_i \tag{58}$$

In view of (32), substitution of (58) into (57) gives

$$[E^b_z(-\underline{M})]_{NR} = \sum_{j=1}^{4} V^e_j \left(\sum_{n=0}^{N2} S^e_{jn}\right) \tag{59}$$

Equating [1, eqs. (8) and (12)] at $\phi = 0$, we obtain

$$\sum_{n=0}^{\infty} S^e_{jn} = \begin{cases} 1 & , \; j=1 \\ 0 & , \; j=2,3,4 \end{cases} \tag{60}$$

If

$$\sum_{n=0}^{N2} S^e_{jn} = \sum_{n=0}^{\infty} S^e_{jn} \tag{61}$$

then (60) would reduce (59) to

$$[E^b_z(-\underline{M})]_{NR} = V^e_1 \tag{62}$$

If NA is odd so that $\phi_L = 0$ when L = (NA+1)/2, then (49) gives

$$[E_z]_L = V^e_1 \ , \quad L = (NA+1)/2 \tag{63}$$

Noting that (62) and (63) have the same right-hand side and recalling that (61) was used to obtain (62), we deduce that any difference between the computed values of the aperture field (49) at the center of the aperture and the interior field (53) at $k\rho_L = ka$ is due to the error in truncating the summation with respect to n in (55), the error in calculating $\{S^e_{jn}\}$, and, of course, roundoff error.

Having presented relevant formulas, we begin to verbally describe the main program. In the equations that were presented in this section, Y^e and Y^o are 4×4 matrices, $\vec{\hat{V}}^e$, $\vec{I}^e$, $\vec{\hat{V}}^o$, and $\vec{I}^o$ are 4×1 column vectors, and $\{V^e_j, V^o_j, j=1,2,3,4\}$ are defined by (45) and (46). The equations that were programmed are those that were presented in this section with the order 4 replaced by NN, that is, Y^e and Y^o are replaced by the $NN \times NN$ submatrices in their upper left-hand corners, and $\hat{V}^e_j$, I^e_j, $\hat{V}^o_j$, I^o_j, V^e_j, and V^o_j are retained only for $j \leq NN$. We describe the main program by relating variables therein to quantities that appear in the previous text of this section. What follows is clarified in Table 1.

Line 27 and DO loop 61 put

$$J_n(0) = \begin{cases} 1 & , \ n = 0 \\ 0 & , \ n > 0 \end{cases} \tag{64}$$

in BJ(n+1) for n=0,1,...,N1-1. Equation (64) can be obtained from [3, formula 9.1.10]. In DO loop 18, line 38 puts $J_n(k\rho_L)$ in BJ(n+1+(L-1)*N1) for n=0,1,...,N1-1 where $k\rho_L$ is given by (22). Line 38 also puts $Y_n(k\rho_L)$ in BY(n+1+(L-1)*N1) for n=0,1,...,N1-1 although only $\{Y_n(k\rho_{NR}) = Y_n(ka)$, n=0,1,...,N1-1$\}$ is needed.

Table 1. Key lines in the main program, the variables read in, defined, or incremented therein, the corresponding quantities in the text, and the numbers of the equations where these quantities appear.

Line Number	Variable(s) in the main program	Quantity in the text	Equation number(s)
22	IP and X	p and ka	16
22	P and ALP	ϕ_o and α in radians	16
27	BJ(1)	$J_o(0)$	55
29	BJ(NP)	$J_{NP-1}(0)$	55
31	XR(1)	$k\rho_1$	22
35	XR(L)	$k\rho_L$	22
38	{BJ(n+JJ), n=0, 1,...,N1-1}	$\{J_n(k\rho_L)$, n=0,1,...,N1-1$\}$	26,30,31,54,55
38	{BY(n+JJ), n=0, 1,...,N1-1}	$\{Y_n(k\rho_L)$, n=0,1,...,N1-1$\}$	26,30
61	{SE(J), J=1,2,3,4}	$\{S^e_{Jp}$, J=1,2,3,4$\}$	32
61	{SO(J), J=1,2,3,4}	$\{S^o_{Jp}$, J=1,2,3,4$\}$	42
62	BJPE	$J_p(ka)$	28
63	CE(1)	C^e_1	32
66	CE(J)	C^e_J	32

Line Number	Variable(s) in the main program	Quantity in the text	Equation Number(s)
69	BJPO	$J_p^o(ka)$ for p=0	36
71	CO(J)	C_J^o for p=0	42
74	BJPO	$J_p^o(ka)$ for p>0	36
75	CO(1)	C_1^o for p>0	42
78	CO(J)	C_J^o for p>0	42
84	N	n	26,30,31
85	{SE(J), J=1,2,3,4}	$\{S_{Jn}^e$, J=1,2,3,4}	31
85	{SO(J), J=1,2,3,4}	$\{S_{Jn}^o$, J=1,2,3,4}	41
85	{FE(I), I=1,2,3,4}	$\{F_{In}^e$, I=1,2,3,4}	26,30
85	{FO(I), I=1,2,3,4}	$\{F_{In}^o$, I=1,2,3,4}	35,40
88	BJN	$J_n(ka)$	26,30
91	U2	$1/H_n^{(2)}(ka)$	26,30
93,94	CSA	$\varepsilon_n \cos(n\alpha)$	26
95	SNA	$2 \sin(n\alpha)$	35
97	EF(I)	$F_{In}^e/H_n^{(2)}(ka)$	30
98	OF(I)	$F_{In}^o/H_n^{(2)}(ka)$	40
99	CUE(I)	nth term	26
100	CUO(I)	nth term	35
105	YE(I)	$S_{1p}^e(EF)_{Ip}$	28
106	YO(I)	$S_{1p}^o(OF)_{Ip}$	38
110	RP(L)	$(RP)_L$	54
116	BJJ(L)	$J_n(k\rho_L)/J_n(ka)$	55
120	BJN	$J_n(ka)\ Y_n(ka)$	31,41
122	EF(I)	jF_{In}^e	30
123	OF(I)	jF_{In}^o	40

Line Number	Variable(s) in the main program	Quantity in the Text	Equation number(s)
125	{BJJ(L), L=1,2,...,NR}	$\{J_n(k\rho_L)/J_n(ka)\}$	55
128,131	E	E_J	67
129,132	O	O_J	68
135	YE(IY)	nth term	28 or 29
136	YO(IY)	nth term	38 or 39
143	RN(JR)	nth term	55
148,152	{VE(J), I=1,2,...,NN}	$\{\hat{V}_I^e,\ I=1,2,...,NN\}$	43
149,154	{VO(I), I=1,2,...,NN}	$\{\hat{V}_I^o,\ I=1,2,...,NN\}$	44
163	VO(1)	V_1^o	46
164	U3	$\hat{V}_1^e$	43
165	VE(1)	V_1^e	45
167	SE(J)	$\|V_J^e\|$	45
168	SO(J)	$\|V_J^o\|$	46
173	SE(J)	$\{S_{J0}^e,\ J=1,2,3,4\}$	47
175	U2	$S_{10}^e\hat{V}_1^e$	47
181	U2	$[E_z^b(-\underline{M})]_{\rho=0}$	47
188	XX	Φ_L/Φ_o	49
192	U2	$[E_z]_L$	49
195	BJ(L)	$\|[E_z]_L\|$	49
201	U2	$(RP)_L\hat{V}_1^e$	53
204	U2	$[E_z^b(-\underline{M})]_L$	53
206	RE(L)	$\|[E_z^b(-\underline{M})]_L\|$	53

Line 61 puts S^e_{Jp} of (32) in SE(J) for J=1,2,3, and 4. Line 61 also puts S^o_{Jp} of (42) in SO(J) for J=1,2,3, and 4. Lines 62 to 67 put $J_p(ka)$ of (28) in BJPE and $\{C^e_J, J=1,2,\ldots,NN\}$ of (32) in {CE(J), J=1,2,...,NN}. Lines 68 to 79 put $J^o_p(ka)$ of (33) in BJPO and $\{C^o_J, J=1,2,\ldots,NN\}$ of (42) in {CO(J), J=1,2,...,NN}.

Being summations with respect to n, the matrix elements I^e_I of (26), Y^e_{IJ} of (28) and (29), I^o_I of (35), Y^o_{IJ} of (38) and (39), and $(RN)_{JL}$ of (55) are accumulated in CUE(I), YE(IY), CUO(I), YO(IY), and RN(JR), respectively, where

$$\left.\begin{aligned} IY &= I + (J-1)*NN \\ JR &= J + (L-1)*NN \end{aligned}\right\} \tag{65}$$

In DO loop 15,

$$N = NP-1 \tag{66}$$

where NP is the index of DO loop 15. The terms for which n=N are added to CUE(I) and CUO(I) in DO loop 15. If $N \neq p$, then the terms for which n=N are added to YE(IY), YO(IY), and RN(JR) in DO loop 15. If N=p, then $S^e_{1p}(EF)_{Ip}$ is added to YE(I), $S^o_{1p}(OF)_{Ip}$ is added to YO(I), and $(RP)_L$ of (54) is put in RP(L) in DO loop 15. More detail on the workings of DO loop 15 is presented in the following four paragraphs.

Line 85 puts S^e_{Jn} of (31) in SE(J), and S^o_{Jn} of (41) in SO(J) for J=1,2,3, and 4. Line 85 also puts F^e_{In} of (26) and (30) in FE(I), and F^o_{In} of (35) and (40) in FO(I) for I=1,2,3 and 4.

If N < N1, then lines 87 to 118 are executed. Line 88 puts $J_n(ka)$ in BJN, line 91 puts $1/H^{(2)}_n(ka)$ in U2, lines 93 and 94 put $\varepsilon_n \cos(n\alpha)$ in CSA, and line 95 puts $2\sin(n\alpha)$ in SNA. In DO loop 40, $(EF)_{In}$ of (30) is

put in EF(I), $(OF)_{In}$ of (40) is put in OF(I), the nth term of (26) is added to CUE(I), and the nth term of (35) is added to CUO(I). If N=p, then DO loops 41 and 53 are executed. In DO loop 41, $S^e_{1p}(EF)_{Ip}$ of (28) is added to YE(I), and $S^o_{1p}(OF)_{Ip}$ of (38) is added to YO(I). In DO loop 53, $(RP)_L$ of (54) is put in RP(L). If N $\neq$ p, then DO loop 55 is executed. In DO loop 55, the factor $J_n(k\rho_L)/J_n(ka)$ in the first summation in (55) is put in BJJ(L).

If N $\geq$ N1, then lines 120 to 125 are executed. Line 120 puts $J_n(ka)\ Y_n(ka)$ of (31) and (41) in BJN. DO loop 51 puts jF^e_{In} of (30) in EF(I) and jF^o_{In} of (40) in OF(I). In line 125, the factor $J_n(k\rho_L)/J_n(ka)$ in the second summation in (55) is put in BJJ(L) for L=1,2,...,NR.

If N $\neq$ p, then lines 126 to 145 are executed. Immediately before these lines are executed $(EF)_{In}$ of (30) will reside in EF(I), and $(OF)_{In}$ of (40) will reside in OF(I). Upon entry into DO loop 44, the quantity E_J defined by

$$E_j = \begin{cases} J_p(ka)\ E^e_{jn}, & j = 1 \\ E^e_{jn} & , j \neq 1 \end{cases} \tag{67}$$

resides in E and the quantity O_J defined by

$$O_j = \begin{cases} J^o_p(ka)\ E^o_{jn}, & j = 1 \\ E^o_{jn} & , j \neq 1 \end{cases} \tag{68}$$

resides in O. Here, E^e_{jn} is given by (31) and E^o_{jn} by (41). In DO loop 44, the nth term of the summation in (28) or (29) is added to YE(IY), and the nth term of the summation in (38) or (39) is added to YO(IY). In nested DO loops 56 and 57, the nth term of the summation in $(RN)_{JL}$ of (55) is added to RN(JR).

Lines 147 to 154 put the elements of the solution $\vec{V}^e$ of (43) in VE and the elements of the solution $\vec{V}^o$ of (44) in VO. In DO loop 45, V_1^e of (45) is accumulated in U2 and V_1^o of (46) is accumulated in U3. The series in (45) and (46) were, as indicated in the paragraph preceding (64), summed for i up to NN rather than 4. After lines 163-165 have been executed, the elements of $\vec{V}^e$ will reside in VE, the elements of $\vec{V}^o$ will reside in VO, and $\hat{V}_1^e$ will be in U3. In DO loop 47, $|V_J^e|$ is put in SE(J), and $|V_J^o|$ is put in SO(J).

The rest of the main program, lines 173 to 210, serves to calculate the field (47) at the center of the cylinder, the aperture field (49), and the field (53) along the line from the center of the cylinder to the center of the aperture. The series in (47), (49), and (53) are summed with i running from 1 to NN rather than from 1 to 4.

Line 173 puts S_{J0}^e in SE(J) for J=1,2,3,4. If p=0, then line 175 puts $S_{10}^e \hat{V}_1^e$ of (47) in U2. If $p \neq 0$, then DO loop 48 is executed and line 181 puts $[E_z^b(-\underline{M})]_{\rho=0}$ of (47) in U2.

The index L of DO loop 62 obtains the subscript L in (49). Line 188 puts (ϕ_L/ϕ_o) in XX. DO loop 63 accumulates $[E_z]_L$ of (49) in U2. Line 195 puts $|[E_z]_L|$ of (49) in BJ(L).

The index L of DO loop 58 obtains the subscript L in (53). Line 201 puts $(RP)_L \hat{V}_1^e$ of (53) in U2. DO loop 59 adds to U2 the summation with respect to j in (53). Line 206 puts $|[E_z^b(-\underline{M})]_L|$ of (53) in RE(L).

Sample input and output data appear after the listing of the main program. The sample input data are for a perfectly conducting circular cylindrical shell of electrical radius ka = 2 with a slot of half angular width $\phi_o = 1.25° = 0.02181662$ radians centered at $\phi = 0°$ illuminated by a TM plane wave that comes from the direction in which $\phi = 0°$. The last

number in the sixth from the last row of the sample output data is exactly the same as the first number in the last row. This is expected because expression (56) is identical to expression (47) when $p=0$.

We ran the main program with IP of the input data changed from 0 to 1. The first four significant figures of each number in the last six lines of the resulting output data were the same as those that were obtained with $IP = 0$. According to (56) and (47), the last number in the sixth from the last line of the output data may, due to roundoff error, be slightly different from the first number in the last line if $IP \neq 0$. These two numbers were identical to each other in the output obtained with $IP = 1$.

Note that the third number in the fourth from the last line of the sample output data is slightly different from the last number in the last row. Reasons for this are given in the paragraph that contains (63).

```
C     LISTING OF THE MAIN PROGRAM
C     INPUT DATA ARE READ FROM THE FILE MAUTZ3.DAT ON UNIT 20.
C     OUTPUT DATA ARE WRITTEN ON THE FILE MAUTZ6.DAT ON UNIT 21.
C     THE SUBROUTINES BES, BESJ, BESJY, BESJJ, SFEO,
C     DECOMP, AND SOLVE ARE USED.
      COMPLEX YE(16),YO(16),CUE(4),CUO(4),U3,U,U2,EF(4),OF(4)
      COMPLEX VE(4),VO(4)
      DIMENSION YO(36),BJ(220),BY(220),XR(11),RP(11),RN(44),RE(11)
      DIMENSION BJJ(11),SE(4),SO(4),FE(4),FO(4),CE(4),CO(4),IPS(4)
      OPEN(UNIT=20,FILE='MAUTZ3.DAT',STATUS='OLD')
      OPEN(UNIT=21,FILE='MAUTZ6.DAT',STATUS='OLD')
      READ(20,25)(YO(I),I=1,36)
   25 FORMAT(5E14.7)
      WRITE(21,26)(YO(I),I=1,36)
   26 FORMAT(' YO'/(1X,5E14.7))
      READ(20,28) N1,N2,N3,NA,NR
   28 FORMAT(I3,I5,3I3)
      WRITE(21,29) N1,N2,N3,NA,NR
   29 FORMAT(' N1=',I3,', N2=',I5,', N3=',I3,', NA=',I3,', NR=',I3)
      N1M=N1-1
      DO 31 JW=1,N3
      READ(20,32) NN,IP,IB,X,P,ALP
   32 FORMAT(3I3,3E14.7)
      WRITE(21,27) NN,IP,IB,X,P,ALP
   27 FORMAT(' NN=',I3,', IP=',I3,', IB=',I3/
     1' X=',E14.7,', P=',E14.7,', ALP=',E14.7)
      BJ(1)=1.
      DO 61 NP=2,N1
      BJ(NP)=0.
   61 CONTINUE
      XR(1)=0.
      JJ=1
      FNR=NR-1
      DO 18 L=2,NR
      XR(L)=(L-1)/FNR*X
      IF(L.EQ.NR) XR(L)=X
      JJ=JJ+N1
      CALL BES(YO,N1M,XR(L),BJ(JJ),BY(JJ))
   18 CONTINUE
      JJN=JJ+N1-1
      WRITE(21,33)(BJ(NP),NP=JJ,JJN)
   33 FORMAT(' BJ'/(1X,5E14.7))
      WRITE(21,19)(BY(NP),NP=JJ,JJN)
   19 FORMAT(' BY'/(1X,5E14.7))
      NN2=NN*NN
      DO 35 I=1,NN2
      YE(I)=0.
      YO(I)=0.
   35 CONTINUE
      DO 36 I=1,NN
      CUE(I)=0.
      CUO(I)=0.
   36 CONTINUE
```

```
      NRN=NR*NN
      DO 52 J=1,NRN
      RN(J)=0.
   52 CONTINUE
      N1R=N1*(NR-1)
      IPP=IP+1+N1R
      IF(IB.EQ.0) BJ(IPP)=0.
      CALL SFEO(IP,P,SE,SO,FE,FO)
      BJPE=BJ(IPP)
      CE(1)=0.
      IF(NN.EQ.1) GO TO 34
      DO 37 J=2,NN
      CE(J)=-SE(J)/SE(1)
   37 CONTINUE
   34 IF(IP.NE.0) GO TO 21
      BJPO=1.
      DO 38 J=1,NN
      CO(J)=0.
   38 CONTINUE
      GO TO 23
   21 BJPO=BJPE
      CO(1)=0.
      IF(NN.EQ.1) GO TO 23
      DO 22 J=2,NN
      CO(J)=-SO(J)/SO(1)
   22 CONTINUE
   23 U3=(1.,0.)
      U=(0.,1.)
      N2P=N2+1
      DO 15 NP=1,N2P
      N=NP-1
      CALL SFEO(N,P,SE,SO,FE,FO)
      IF(N.GE.N1) GO TO 50
      NPN=NP+N1R
      BJN=BJ(NPN)
      S1=BJN
      IF(ABS(S1).LE.1.E-10) S1=0.
      U2=1./CMPLX(S1,-BY(NPN))
      ALN=ALP*N
      CSA=2.*COS(ALN)
      IF(N.EQ.0) CSA=1.
      SNA=2.*SIN(ALN)
      DO 40 I=1,NN
      EF(I)=FE(I)*U2
      OF(I)=FO(I)*U2
      CUE(I)=CSA*EF(I)*U3+CUE(I)
      CUO(I)=SNA*OF(I)*U3+CUO(I)
   40 CONTINUE
      U3=U3*U
      IF(N.NE.IP) GO TO 20
      DO 41 I=1,NN
      YE(I)=SE(1)*EF(I)+YE(I)
      YO(I)=SO(1)*OF(I)+YO(I)
```

```
   41 CONTINUE
      IJ=NP
      DO 53 L=1,NR
      RP(L)=SE(1)*BJ(IJ)
      IJ=IJ+N1
   53 CONTINUE
      GO TO 15
   20 JJ=NP
      DO 55 L=1,NR
      BJJ(L)=BJ(JJ)/BJN
      JJ=JJ+N1
   55 CONTINUE
      GO TO 54
   50 CALL BESJY(N,X,BJN)
      DO 51 I=1,NN
      EF(I)=FE(I)*U
      OF(I)=FO(I)*U
   51 CONTINUE
      CALL BESJJ(N,NR,XR,BJJ)
   54 IY=0
      DO 42 J=1,NN
      E=(SE(J)+CE(J)*SE(1))/BJN
      O=(SO(J)+CO(J)*SO(1))/BJN
      IF(J.NE.1) GO TO 43
      E=BJPE*E
      O=BJPO*O
   43 DO 44 I=1,NN
      IY=IY+1
      YE(IY)=E*EF(I)+YE(IY)
      YO(IY)=O*OF(I)+YO(IY)
   44 CONTINUE
   42 CONTINUE
      JR=0
      DO 56 L=1,NR
      DO 57 J=1,NN
      JR=JR+1
      RN(JR)=BJJ(L)*SE(J)+RN(JR)
   57 CONTINUE
   56 CONTINUE
   15 CONTINUE
      IF(NN.NE.1) GO TO 17
      VE(1)=CUE(1)/YE(1)
      VO(1)=CUO(1)/YO(1)
      GO TO 16
   17 CALL DECOMP(NN,IPS,YE)
      CALL SOLVE(NN,IPS,YE,CUE,VE)
      CALL DECOMP(NN,IPS,YO)
      CALL SOLVE(NN,IPS,YO,CUO,VO)
   16 CE(1)=BJPE
      CO(1)=BJPO
      U2=0.
      U3=0.
      DO 45 I=1,NN
```

```
      U2=CE(I)*VE(I)+U2
      U3=CO(I)*VO(I)+U3
   45 CONTINUE
      VO(1)=U3
      U3=VE(1)
      VE(1)=U2
      DO 47 J=1,NN
      SE(J)=CABS(VE(J))
      SO(J)=CABS(VO(J))
   47 CONTINUE
      WRITE(21,24)(SE(J),J=1,NN)
      WRITE(21,24)(SO(J),J=1,NN)
   24 FORMAT(1X,4E14.7)
      CALL SFEO(O,P,SE,SO,FE,FO)
      IF(IP.NE.0) GO TO 46
      U2=SE(1)*U3
      GO TO 39
   46 U2=0.
      DO 48 J=1,NN
      U2=SE(J)*VE(J)+U2
   48 CONTINUE
      U2=U2/BJ(N1R+1)
   39 S1=CABS(U2)
      WRITE(21,49) U2,S1
   49 FORMAT(' E=',2E14.7,', ABS(E)=',E14.7)
      FNA=2./(NA-1)
      DO 62 L=1,NA
      U2=0.
      XX=-1.+(L-1)*FNA
      X2=XX*XX
      S1=SQRT(1.-X2)
      DO 63 J=1,NN
      U2=S1*(XX*VO(J)+VE(J))+U2
      S1=S1*X2
   63 CONTINUE
      BJ(L)=CABS(U2)
   62 CONTINUE
      WRITE(21,64)(BJ(L),L=1,NA)
   64 FORMAT(' APERTURE FIELD AMPLITUDE'/(1X,4E14.7))
      JR=0
      DO 58 L=1,NR
      U2=RP(L)*U3
      DO 59 J=1,NN
      JR=JR+1
      U2=RN(JR)*VE(J)+U2
   59 CONTINUE
      RE(L)=CABS(U2)
   58 CONTINUE
      WRITE(21,60)(RE(L),L=1,NR)
   60 FORMAT(' INTERIOR FIELD AMPLITUDE'/(1X,4E14.7))
   31 CONTINUE
      STOP
      END
```

```
C
C     LISTING OF THE INPUT DATA FILE MAUTZ3.DAT
-0.3072582E+04 0.7368758E+04-0.6085100E+03 0.1710234E+02-0.2271001E+00
 0.1600171E-02-0.5961089E-05 0.9545773E-08 0.4163150E+05 0.3420211E+03
 0.1000000E+01-0.6024727E+04 0.1613512E+04-0.7532210E+02 0.1402590E+01
-0.1275602E-01 0.5832787E-04-0.1107698E-06 0.3072946E+05 0.2886431E+03
 0.1000000E+01 0.9999999E+00-0.1097659E-02 0.2461455E-04 0.1000000E+01
 0.1829893E-02-0.3191328E-04-0.1562498E-01 0.1427079E-03-0.5937434E-05
 0.4687498E-01-0.1998720E-03 0.7317495E-05 0.6366198E+00 0.7853982E+00
 0.2356194E+01
 2010000  1  5  3
  4  0  1 0.2000000E+01 0.2181662E-01 0.0000000E+00
C
C     LISTING OF THE OUTPUT DATA FILE MAUTZ6.DAT
 YO
 -0.3072582E+04 0.7368758E+04-0.6085100E+03 0.1710234E+02-0.2271001E+00
  0.1600171E-02-0.5961089E-05 0.9545773E-08 0.4163150E+05 0.3420211E+03
  0.1000000E+01-0.6024727E+04 0.1613512E+04-0.7532210E+02 0.1402590E+01
 -0.1275602E-01 0.5832787E-04-0.1107698E-06 0.3072946E+05 0.2886431E+03
  0.1000000E+01 0.9999999E+00-0.1097659E-02 0.2461455E-04 0.1000000E+01
  0.1829893E-02-0.3191328E-04-0.1562498E-01 0.1427079E-03-0.5937434E-05
  0.4687498E-01-0.1998720E-03 0.7317495E-05 0.6366198E+00 0.7853982E+00
  0.2356194E+01
 N1= 20, N2=10000, N3=  1, NA=  5, NR=  3
 NN=  4, IP=  0, IB=  1
 X= 0.2000000E+01, P= 0.2181662E-01, ALP= 0.0000000E+00
 BJ
  0.2238908E+00 0.5767248E+00 0.3528340E+00 0.1289432E+00 0.3399571E-01
  0.7039629E-02 0.1202429E-02 0.1749441E-03 0.2217955E-04 0.2492343E-05
  0.2515386E-06 0.2304285E-07 0.1932695E-08 0.1494942E-09 0.1072946E-10
  0.7183016E-12 0.4506005E-13 0.2659308E-14 0.1481737E-15 0.7819242E-17
 BY
  0.5103757E+00-0.1070324E+00-0.6174081E+00-0.1127784E+01-0.2765943E+01
 -0.9935989E+01-0.4691401E+02-0.2715481E+03-0.1853922E+04-0.1455983E+05
 -0.1291846E+06-0.1277286E+07-0.1392096E+08-0.1657742E+09-0.2141144E+10
 -0.2981024E+11-0.4450125E+12-0.7090389E+13-0.1200916E+15-0.2154558E+16
  0.4655501E-01 0.1714323E-01 0.6358480E-01 0.5726649E-01
  0.0000000E+00 0.0000000E+00 0.0000000E+00 0.0000000E+00
 E=-0.1121177E-02-0.2728021E-03, ABS(E)= 0.1153889E-02
 APERTURE FIELD AMPLITUDE
  0.0000000E+00 0.4136272E-01 0.4655501E-01 0.4136272E-01
  0.0000000E+00
 INTERIOR FIELD AMPLITUDE
  0.1153889E-02 0.1597396E-02 0.4656652E-01
```

REFERENCES

[1] J. R. Mautz and R. F. Harrington, "Electromagnetic Penetration into a Conducting Circular Cylinder through a Narrow Slot, TM Case," Technical Report SYRU/DECE/TR-86/4, Department of Electrical and Computer Engineering, Syracuse University, Syracuse, NY 13244-1240, Dec. 1986.

[2] J. R. Mautz and R. F. Harrington, "Computer Programs for Electromagnetic Scattering from a Slotted TM Cylindrical Conductor by the Pseudo-Image Method," Technical Report SYRU/DECE/TR-85/5, Department of Electrical and Computer Engineering, Syracuse University, Syracuse, NY 13244-1240, Nov. 1985.

[3] M. Abramowitz and I. A. Stegun, Handbook of Mathematical Functions, U. S. Government Printing Office, Washington, D.C. (Natl. Bur. Std. U.S. Applied Math. Ser. 55), 1964.

[4] H. B. Dwight, Tables of Integrals and Other Mathematical Data, Macmillan, New York, 1961.

[5] J. R. Mautz and R. F. Harrington, "Transmission from a Rectangular Waveguide into Half Space through a Rectangular Aperture," Technical Report TR-76-5, Department of Electrical and Computer Engineering, Syracuse University, Syracuse, NY 13244-1240, May 1976.

END

4-87

DTIC